

# **Producing Reports with SAS**

**by Elvira Agrón**

TASC/Advanced Support Team  
Center for Information Technology  
National Institutes of Health

May 17-18, 2001



# Table of Contents

Section	Page
<u>Course Description</u>	4
<u>Documentation</u>	5
<u>The Sample Data Set: AWARDS</u>	6
<u>Usage Modes</u>	9
<u>Sample Reports</u>	10
<u>The PROC REPORT Statement</u>	19
<u>Variables to Include in Report</u>	22
<u>Usage of Variable</u>	23
<u>The DEFINE Statement</u>	24
<u>Display Variable</u>	26
<u>Order Variable</u>	29
<u>Analysis Variable</u>	32
<u>The BREAK Statement</u>	36
<u>The RBREAK Statement</u>	38
<u>Group Variable</u>	41
<u>Multiple Statistics</u>	43
<u>Workshop 1</u>	46
<u>Across Variable</u>	53
<u>Computed Variable</u>	57
<u>Compute Blocks</u>	64
<u>LINE Statement</u>	74
<u>Workshop 2</u>	75
<u>How a Report is Built</u>	79
<u>Missing Data</u>	85
<u>Modifying the Data to Get the Report you Want</u>	88
<u>Multiple Panels per Page</u>	92
<u>Displaying Computed Variable Before Variables it Uses</u>	94
<u>Workshop 3</u>	97
<u>Saving a Report Definition</u>	101
<u>Using a Report Definition</u>	103
<u>Saving the Results in a SAS Data Set</u>	105
<u>The PAGE Option</u>	107
<u>Output Delivery System</u>	111
<u>Workshop 4</u>	114
<u>Appendix: Workshop Answers</u>	116

# Course Description

This course shows you how to create reports using the REPORT procedure of the Base SAS System:

Specifically, you will learn how to:

- create various styles of reports
- add summarization lines
- include statistics in the report
- calculate new columns
- customize the report
- create an output SAS data set
- save the report definition permanently
- use the report definition

These course notes can be found in the Training section of our website: <http://statsoft.nih.gov>.

# Documentation

- SAS Guide to the REPORT Procedure, Release 6.11
- SAS Guide to the REPORT Procedure, Usage and Reference, Version 6
- SAS Technical Report P-258: Using the REPORT Procedure in a Nonwindowing Environment, Release 6.07

View the SAS OnlineDoc, Version 8, in our website:

<http://statsoft.nih.gov>

# The Sample Data Set: **AWARDS**

The SAS data set **AWARDS** contains information on grants given to medical schools in DC, MD and VA in 1996 and 1997. There is an observation for each school with the total number of awards.

The variables are:

<b>YEAR</b>	year award granted
<b>STNAME</b>	state name
<b>UNIVNAME</b>	university name
<b>CONN</b>	number of contracts awards
<b>CONAMT</b>	amount of contracts awards

# The Sample Data Set: AWARDS

<b>FELN</b>	number of fellowship awards
<b>FELAMT</b>	amount of fellowship awards
<b>RESN</b>	number of research awards
<b>RESAMT</b>	amount of research awards
<b>TRAN</b>	number of training awards
<b>TRAAMT</b>	amount of training awards
<b>OTHN</b>	number of other awards
<b>OTHAMT</b>	amount of other awards

# The Sample Data Set: A W A R D S

OBS	UNI VNAME	STNAME	RESN	RESAMT
1	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	MARYLAND	560	181608047
2	UNIVERSITY OF MARYLAND SCH OF MEDICINE	MARYLAND	221	47330824
3	UNIVERSITY OF VIRGINIA SCH OF MEDICINE	VIRGINIA	178	41466312
4	GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	DIST OF COL	134	32118502
5	VIRGINIA COMMONWEALTH UNIV MED COL OF VA	VIRGINIA	141	33955259
6	GEORGE WASHINGTON UNIV SCH OF MEDICINE	DIST OF COL	26	7097448
7	HOWARD UNIVERSITY COLLEGE OF MEDICINE	DIST OF COL	14	4904997
8	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	VIRGINIA	18	3472404
9	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	MARYLAND	579	187473322
10	UNIVERSITY OF MARYLAND SCH OF MEDICINE	MARYLAND	237	50614013
11	UNIVERSITY OF VIRGINIA SCH OF MEDICINE	VIRGINIA	208	47942379
12	GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	DIST OF COL	140	33651458
13	VIRGINIA COMMONWEALTH UNIV MED COL OF VA	VIRGINIA	133	34256198
14	HOWARD UNIVERSITY COLLEGE OF MEDICINE	DIST OF COL	15	6569248
15	GEORGE WASHINGTON UNIV SCH OF MEDICINE	DIST OF COL	27	7015966
16	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	VIRGINIA	16	3248901
17	U S UNIFORMED SERVICES UNIV HLTH SCI S	MARYLAND	1	97773

OBS	TRAN	TRAAMT	CONN	CONAMT	FELN	FELAMT	OTHN	OTHAMT	YEAR
1	44	8795463	17	10304633	57	1624258	3	1574112	1996
2	10	878818	8	3945118	11	276044	0	0	1996
3	14	2442739	0	0	22	622612	1	132978	1996
4	4	461360	11	5288387	10	246684	0	0	1996
5	10	1253788	5	1647325	9	221020	0	0	1996
6	0	0	2	1204952	6	136783	1	734544	1996
7	2	623425	1	75000	2	70948	0	0	1996
8	0	0	1	2670	2	48626	0	0	1996
9	42	9478438	15	11788806	50	1476525	1	86085	1997
10	12	1369113	6	2591474	13	360007	1	148886	1997
11	14	2783142	0	0	22	618832	2	344422	1997
12	6	884139	8	4024313	14	340851	0	0	1997
13	6	1007380	5	1515404	8	233707	0	0	1997
14	2	332227	2	1892193	1	30289	0	0	1997
15	0	0	3	1254142	4	126803	0	0	1997
16	0	0	1	31178	1	16718	0	0	1997
17	0	0	0	0	0	0	1	149990	1997



# Usage Modes

The REPORT procedure can be used in two modes:

- windowing environment
  - a “point-and-click” facility
- non-windowing environment
  - you submit statements to SAS

In this course we will discuss the non-windowing environment.

# **Sample Reports**

# Sample 1

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	University	Number Awards	Total Amount
1996	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	560	\$181,608,047
1996	MARYLAND	UNIVERSITY OF MARYLAND SCH OF MEDICINE	221	\$47,330,824
1996	VIRGINIA	UNIVERSITY OF VIRGINIA SCH OF MEDICINE	178	\$41,466,312
1996	DIST OF COL	GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	134	\$32,118,502
1996	VIRGINIA	VIRGINIA COMMONWEALTH UNIV MED COL OF VA	141	\$33,955,259
1996	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	26	\$7,097,448
1996	DIST OF COL	HOWARD UNIVERSITY COLLEGE OF MEDICINE	14	\$4,904,997
1996	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	18	\$3,472,404
1997	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	579	\$187,473,322
1997	MARYLAND	UNIVERSITY OF MARYLAND SCH OF MEDICINE	237	\$50,614,013
1997	VIRGINIA	UNIVERSITY OF VIRGINIA SCH OF MEDICINE	208	\$47,942,379
1997	DIST OF COL	GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	140	\$33,651,458
1997	VIRGINIA	VIRGINIA COMMONWEALTH UNIV MED COL OF VA	133	\$34,256,198
1997	DIST OF COL	HOWARD UNIVERSITY COLLEGE OF MEDICINE	15	\$6,569,248
1997	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	27	\$7,015,966
1997	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	16	\$3,248,901
1997	MARYLAND	U S UNIFORMED SERVICES UNIV HLTH SCI S	1	\$97,773

# Sample 2

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	Uni versi ty	Number Awards	Total Amount
1996	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	26	\$7,097,448
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	134	\$32,118,502
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	14	\$4,904,997
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	560	\$181,608,047
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	221	\$47,330,824
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	18	\$3,472,404
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	178	\$41,466,312
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	141	\$33,955,259
1997	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	27	\$7,015,966
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	140	\$33,651,458
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	15	\$6,569,248
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	579	\$187,473,322
		U S UNIFORMED SERVICES UNIV HLTH SCI S	1	\$97,773
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	237	\$50,614,013
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	16	\$3,248,901
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	208	\$47,942,379
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	133	\$34,256,198

# Sample 3

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

			Research Grants	
Year	State	University	N	Total
1996	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	26	\$7,097,448
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	134	\$32,118,502
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	14	\$4,904,997
			174	\$44,120,947
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	560	\$181,608,047
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	221	\$47,330,824
			781	\$228,938,871
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	18	\$3,472,404
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	178	\$41,466,312
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	141	\$33,955,259
			337	\$78,893,975
			1,292	\$351,953,793

# Sample 3

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	University	Research Grants	
			N	Total
1997	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	27	\$7,015,966
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	140	\$33,651,458
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	15	\$6,569,248
			182	\$47,236,672
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	579	\$187,473,322
		U S UNIFORMED SERVICES UNIV HLTH SCI S	1	\$97,773
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	237	\$50,614,013
			817	\$238,185,108
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	16	\$3,248,901
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	208	\$47,942,379
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	133	\$34,256,198
			357	\$85,447,478
			1,356	\$370,869,258

# Sample 4

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	Number Awards	Total Amount
ff			
1996	DIST OF COL	174	\$44,120,947
	MARYLAND	781	\$228,938,871
	VIRGINIA	337	\$78,893,975
1997	DIST OF COL	182	\$47,236,672
	MARYLAND	817	\$238,185,108
	VIRGINIA	357	\$85,447,478

# Sample 5

NIH Research Awards Granted to Medical Schools  
 District of Columbia, Maryland and Virginia  
 Minimum and Maximum Amounts

Year	State	Minimum	Maximum
1996	DIST OF COL	\$4,904,997	\$32,118,502
	MARYLAND	\$47,330,824	\$181,608,047
	VIRGINIA	\$3,472,404	\$41,466,312
1997	DIST OF COL	\$6,569,248	\$33,651,458
	MARYLAND	\$97,773	\$187,473,322
	VIRGINIA	\$3,248,901	\$47,942,379



# Sample 6

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

	_____1996_____		_____1997_____	
State	Number Awards	Total Amount	Number Awards	Total Amount
DI ST OF COL	174	\$44, 120, 947	182	\$47, 236, 672
MARYLAND	781	\$228, 938, 871	817	\$238, 185, 108
VI RGI NI A	337	\$78, 893, 975	357	\$85, 447, 478

# Sample 7

## NIH Awards Granted to Medical Schools District of Columbia, Maryland and Virginia

Year	State	Contracts	Fellowships	Research	Training	Other	Total
1996	DIST OF COL	14	18	174	6	1	213
	MARYLAND	25	68	781	54	3	931
	VIRGINIA	6	33	337	24	1	401
		ffffffffffff	ffffffffffff	ffffffffffff	ffffffffffff	ffffffffffff	ffffffffffff
		45	119	1,292	84	5	1,545
1997	DIST OF COL	13	19	182	8	0	222
	MARYLAND	21	63	817	54	3	958
	VIRGINIA	6	31	357	20	2	416
		ffffffffffff	ffffffffffff	ffffffffffff	ffffffffffff	ffffffffffff	ffffffffffff
		40	113	1,356	82	5	1,596

# The PROC REPORT Statement

This statement invokes the procedure.

**PROC REPORT** *options*;

Some options are:

DATA=	data set to process
OUT=	output data set to create
NOWD	invokes the non-windowing mode
HEADSKIP	writes a blank line under column headers
HEADLINE	writes a line under the header
MISSING	treats missing values as valid values for order, group or across variables
SPLIT=	specifies the splitting character used in the column headers (default: /)

# The PROC REPORT Statement

PANELS=	specifies the number of panels of the report
COLWIDTH=	specifies the column width for all columns containing computed or numeric data set variables (default: 9)
OUTREPT=	specifies the catalog entry where the report definition should be saved
REPORT=	specifies the report definition to use
NOEXEC	suppresses the building of the report

## New in Version 8:

FORMCHAR=	defines the characters used to draw lines and boxes in the report
BOX	surrounds report items in boxes (like PROC FREQ)

# The PROC REPORT Statement

## Examples:

```
proc report data=rep.awards nowd headski p; 
```

```
proc report data=rep.awards nowd headski p headl i ne; 
```

```
proc report data=rep.awards nowd  
    outrept=rep.reports.report1 noexec;
```

# Variables to Include in Report

Specify the variables you want to display or use in the report with the **COLUMN** statement. The columns will be displayed in the order you list them. The COLUMN statement is not required.

## Examples:

```
column year stname uni vname resn resamt;
```



```
column year stname uni vname  
      ('Research Grants' ' --' resn resamt);
```



```
column stname year, (resn resamt);
```



```
column year stname resamt=resmin resamt=resmax;
```



# Usage of Variable

PROC REPORT uses variables according to the usages you specify in the **DEFINE** statements. Valid usages are:

DISPLAY

ORDER

GROUP

ACROSS

ANALYSIS

COMPUTED

By default, character variables are defined as display variables and numeric variables are defined as analysis variables.

# The DEFINE Statement

**DEFINE** *var / usage options;*

*usage* defines the usage type

*options* this is a partial list of the options

FORMAT= format to use with the column (abb. F=)

DESCENDING displays values in descending order

ORDER= orders values

DATA by their order in data set

FORMATTED by the formatted values  
(default)

FREQ by ascending count

INTERNAL by their unformatted values



# The DEFINE Statement

<i>column_header</i>	column label in quotes
CENTER LEFT RIGHT	specifies the justification of the formatted values within the column width (default: numeric/right, character/left)
SPACING=	number of blank characters to leave between the column being defined and the one to its left (default: 2)
WIDTH=	defines the column width
NOPRINT	suppresses the display of the column
<i>statistic</i>	see <b>Analysis Variable</b> section
DEFINE statements are optional.	

# Display Variable

They appear as they do in the input data set.

When one or more display variable appears in the report, every observation of the data set is displayed in the report.

# Display Variable

In **sample 1** all variables are defined as display variables:

```
title 'NIH Research Awards Granted to Medical Schools';
title2 'District of Columbia, Maryland and Virginia';
proc report data=rep.awards nowd headskip;
  column year stname uni vname resn resamt;
  define year / 'Year' display;
  define stname / 'State' display;
  define uni vname / 'University' display;
  define resn / 'Number/Awards' display;
  define resamt / 'Total/Amount' display;
  format resamt dollar14.;
run;
```

Note: The slash in 'Number/Awards' is a split character.

# Sample 1

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	University	Number Awards	Total Amount
1996	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	560	\$181,608,047
1996	MARYLAND	UNIVERSITY OF MARYLAND SCH OF MEDICINE	221	\$47,330,824
1996	VIRGINIA	UNIVERSITY OF VIRGINIA SCH OF MEDICINE	178	\$41,466,312
1996	DIST OF COL	GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	134	\$32,118,502
1996	VIRGINIA	VIRGINIA COMMONWEALTH UNIV MED COL OF VA	141	\$33,955,259
1996	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	26	\$7,097,448
1996	DIST OF COL	HOWARD UNIVERSITY COLLEGE OF MEDICINE	14	\$4,904,997
1996	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	18	\$3,472,404
1997	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	579	\$187,473,322
1997	MARYLAND	UNIVERSITY OF MARYLAND SCH OF MEDICINE	237	\$50,614,013
1997	VIRGINIA	UNIVERSITY OF VIRGINIA SCH OF MEDICINE	208	\$47,942,379
1997	DIST OF COL	GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	140	\$33,651,458
1997	VIRGINIA	VIRGINIA COMMONWEALTH UNIV MED COL OF VA	133	\$34,256,198
1997	DIST OF COL	HOWARD UNIVERSITY COLLEGE OF MEDICINE	15	\$6,569,248
1997	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	27	\$7,015,966
1997	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	16	\$3,248,901
1997	MARYLAND	U S UNIFORMED SERVICES UNIV HLTH SCI S	1	\$97,773

# Order Variable

Rows in the report are sorted according to the order variables when there is no display variable to the left of an order variable.

By default, rows are sorted by the formatted values of the order variables if they are formatted. If not, then the rows are sorted by the internal values.

The option DESCENDING can be used in the DEFINE statement to reverse the order of the rows.

# Order Variable

In **sample 2** YEAR, STNAME and UNIVNAME are order variables:

```
proc report data=rep.awards nowd headskip headline;  
  column year stname univname resn resamt;  
  define year / 'Year' order;  
  define stname / 'State' order;  
  define univname / 'University' order;  
  define resn / 'Number/Awards' display;  
  define resamt / 'Total/Amount' display;  
  format resamt dollar14. ;  
run;
```

# Sample 2

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	Uni versi ty	Number Awards	Total Amount
1996	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	26	\$7,097,448
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	134	\$32,118,502
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	14	\$4,904,997
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	560	\$181,608,047
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	221	\$47,330,824
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	18	\$3,472,404
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	178	\$41,466,312
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	141	\$33,955,259
1997	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	27	\$7,015,966
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	140	\$33,651,458
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	15	\$6,569,248
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	579	\$187,473,322
		U S UNIFORMED SERVICES UNIV HLTH SCI S	1	\$97,773
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	237	\$50,614,013
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	16	\$3,248,901
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	208	\$47,942,379
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	133	\$34,256,198

# Analysis Variable

An analysis variable is used to calculate statistics for all the observations represented by a group of observations.

By default, numeric variables are treated as analysis variables used to calculate the SUM statistic. Valid statistics are:

N	NMISS	MEAN	STD	MIN	MAX
RANGE	SUM	USS	CSS	STDERR	CV
T	PRT	VAR	SUMWGT	PCTN	PCTSUM

New in 8.1:

P1	P5	P10	P25/Q1	P50/Q2/MEDIAN
P75/Q3	P90	P95	P99	QRANGE



# Analysis Variable

In **sample 3** RESN and RESAMT are defined as analysis variables so we can get values in the summary line.

```
proc report data=rep.awards nowd headskip headline;  
  column year stname uni vname  
           (' Research Grants' ' --' resn resamt);  
  define year / 'Year' order;  
  define stname / 'State' order;  
  define uni vname / 'Uni versi ty' order;  
  define resn / 'N' analysis sum;  
  define resamt / 'Total' analysis sum;  
  break after stname / skip summarize ol suppress;  
  break after year / page summarize dul dol suppress;  
  format resn comma5. resamt dollar14. ;  
run;
```

Note: The BREAK statement will be covered in the next section.

# Sample 3

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

			Research Grants	
Year	State	University	N	Total
1996	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	26	\$7,097,448
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	134	\$32,118,502
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	14	\$4,904,997
			174	\$44,120,947
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	560	\$181,608,047
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	221	\$47,330,824
			781	\$228,938,871
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	18	\$3,472,404
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	178	\$41,466,312
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	141	\$33,955,259
			337	\$78,893,975
			1,292	\$351,953,793

# Sample 3

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

			Research Grants	
Year	State	University	N	Total
1997	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	27	\$7,015,966
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	140	\$33,651,458
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	15	\$6,569,248
			182	\$47,236,672
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	579	\$187,473,322
		U S UNIFORMED SERVICES UNIV HLTH SCI S	1	\$97,773
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	237	\$50,614,013
			817	\$238,185,108
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	16	\$3,248,901
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	208	\$47,942,379
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	133	\$34,256,198
			357	\$85,447,478
			1,356	\$370,869,258

# The BREAK Statement

Produces a summary line when the value of an order or group variable changes.

**BREAK** *location var / options;*

*location*      placement of break lines: AFTER, BEFORE

*var*              order or group variable

*options*        this is a partial list of the options

DOL    for double overlining

DUL    for double underlining

OL      for single overlining

UL      for single underlining

# The BREAK Statement

SKIP	skips a line after each group
PAGE	starts a new page after each group
SUMMARIZE	writes a summary line for each group
SUPPRESS	suppresses value and lines of the break variable in summary line

**Examples:** In **sample 3** we use two BREAK statements

```
break after stname / skip summarize ol suppress;  
break after year / page;
```

# The RBREAK Statement

Produces a summary line at the beginning or end of a report.

**RBREAK** *location / options;*

*location*      placement of break lines: AFTER, BEFORE

*options*      this is a partial list of the options

DOL              for double overlining

DUL              for double underlining

OL                for overlining

UL                for underlining

SKIP             skips a line after a break line  
located at beginning of the report

SUMMARIZE      writes summary line

# The RBREAK Statement

## Example:

```
proc report data=rep.awards nowd headskip headline;  
  column year stname resn resamt;  
  define year / 'Year' group;  
  define stname / 'State' group;  
  define resn / 'Number/Awards' analysis sum f=comma6. ;  
  define resamt / 'Total/Amount' analysis sum f=dollar13. ;  
  break after year / skip summarize ol ;  
  rbreak before / summarize dul skip;  
run;
```

# The RBREAK Statement

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	Number Awards	Total Amount
ff			
		2,648	\$722,823,051
		=====	=====
1996	DIST OF COL	174	\$44,120,947
	MARYLAND	781	\$228,938,871
	VIRGINIA	337	\$78,893,975
fffffffff		ffffff	ffffffffffffff
1996		1,292	\$351,953,793
1997	DIST OF COL	182	\$47,236,672
	MARYLAND	817	\$238,185,108
	VIRGINIA	357	\$85,447,478
fffffffff		ffffff	ffffffffffffff
1997		1,356	\$370,869,258



# Group Variable

Group variables are used to consolidate into one row all observations that have a unique combination of values.

In **sample 4** YEAR and STNAME are group variables.

```
proc report data=rep.awards nowd headskip headline;  
  column year stname resn resamt;  
  define year / 'Year' group;  
  define stname / 'State' group;  
  define resn / 'Number/Awards' analysis sum;  
  define resamt / 'Total/Amount' analysis sum;  
  break after year / skip;  
  format resamt dollar13. ;  
run;
```

# Sample 4

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	Number Awards	Total Amount
<i>ff</i>			
1996	DIST OF COL	174	\$44,120,947
	MARYLAND	781	\$228,938,871
	VIRGINIA	337	\$78,893,975
1997	DIST OF COL	182	\$47,236,672
	MARYLAND	817	\$238,185,108
	VIRGINIA	357	\$85,447,478

# Multiple Statistics

There are various ways of requesting multiple statistics for the same group variable.

## Example 1:

```
proc report data=rep.awards nowd headskip headline;  
  column year sname resamt, (min max);  
  define year / 'Year' group;  
  define sname / 'State' group;  
  define resamt / ' ';  
  define min / 'Minimum' f=dollar12.;  
  define max / 'Maximum' f=dollar12.;  
  break after year / skip;  
run;
```

# Multiple Statistics

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia  
Minimum and Maximum Any University Received

Year	State	Minimum	Maximum
1996	DIST OF COL	\$4,904,997	\$32,118,502
	MARYLAND	\$47,330,824	\$181,608,047
	VIRGINIA	\$3,472,404	\$41,466,312
1997	DIST OF COL	\$6,569,248	\$33,651,458
	MARYLAND	\$97,773	\$187,473,322
	VIRGINIA	\$3,248,901	\$47,942,379

# Multiple Statistics

**Example 2: Sample 5** produces the same results.

```
proc report data=rep.awards nowd headskip headline;  
  column year stname resamt=resmin resamt=resmax;  
  define year / 'Year' group;  
  define stname / 'State' group;  
  define resmin / 'Minimum' analysis min f=dollar12.;  
  define resmax / 'Maximum' analysis max f=dollar12.;  
  break after year / skip;  
run;
```

# **Workshop 1**

For all the exercises in all the workshops use the data set AWARDS that is saved in the folder `c:\awards`.

Before beginning the exercises enter and submit the following statements:

```
libname rep 'c:\awards' ;
```

You may want to save your programs in the folder `c:\awards` since you will need to modify some later.

# Exercise 1:

Create the report shown below. All observations should appear in the report.

NIH Fellowship, Research and Training Awards					
Year	State	University	Fellowship	Research	Training
fff					
1996	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	\$136,783	\$7,097,448	\$0
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	\$246,684	\$32,118,502	\$461,360
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	\$70,948	\$4,904,997	\$623,425
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	\$1,624,258	\$181,608,047	\$8,795,463
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	\$276,044	\$47,330,824	\$878,818
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	\$48,626	\$3,472,404	\$0
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	\$622,612	\$41,466,312	\$2,442,739
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	\$221,020	\$33,955,259	\$1,253,788
1997	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	\$126,803	\$7,015,966	\$0
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	\$340,851	\$33,651,458	\$884,139
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	\$30,289	\$6,569,248	\$332,227
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	\$1,476,525	\$187,473,322	\$9,478,438
		U S UNIFORMED SERVICES UNIV HLTH SCIS	\$0	\$97,773	\$0
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	\$360,007	\$50,614,013	\$1,369,113
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	\$16,718	\$3,248,901	\$0
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	\$618,832	\$47,942,379	\$2,783,142
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	\$233,707	\$34,256,198	\$1,007,380



## Exercise 2:

Modify the program from exercise 1 as follows:

- each year should be in a separate page
- include a summary line after each state and after each year

This is the report for 1996:

NIH Fellowship, Research and Training Awards					
Year	State	University	Fellowship	Research	Training
%%%					
1996	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	\$136,783	\$7,097,448	\$0
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	\$246,684	\$32,118,502	\$461,360
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	\$70,948	\$4,904,997	\$623,425
			ffffffffffffff	ffffffffffffff	ffffffffffffff
			\$454,415	\$44,120,947	\$1,084,785
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	\$1,624,258	\$181,608,047	\$8,795,463
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	\$276,044	\$47,330,824	\$878,818
			ffffffffffffff	ffffffffffffff	ffffffffffffff
			\$1,900,302	\$228,938,871	\$9,674,281
VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	\$48,626	\$3,472,404	\$0	
	UNIVERSITY OF VIRGINIA SCH OF MEDICINE	\$622,612	\$41,466,312	\$2,442,739	
	VIRGINIA COMMONWEALTH UNIV MED COL OF VA	\$221,020	\$33,955,259	\$1,253,788	
		ffffffffffffff	ffffffffffffff	ffffffffffffff	
		\$892,258	\$78,893,975	\$3,696,527	
		=====	=====	=====	
		\$3,246,975	\$351,953,793	\$14,455,593	
		-----	-----	-----	

## Exercise 3:

Create the following report. It is similar to exercise 2 but only includes the summary lines and excludes the university names.

NIH Fellowship, Research and Training Awards				
Year	State	Fellowship	Research	Training
1996	DIST OF COL	\$454,415	\$44,120,947	\$1,084,785
	MARYLAND	\$1,900,302	\$228,938,871	\$9,674,281
	VIRGINIA	\$892,258	\$78,893,975	\$3,696,527
		=====	=====	=====
		\$3,246,975	\$351,953,793	\$14,455,593
1997	DIST OF COL	\$497,943	\$47,236,672	\$1,216,366
	MARYLAND	\$1,836,532	\$238,185,108	\$10,847,551
	VIRGINIA	\$869,257	\$85,447,478	\$3,790,522
		=====	=====	=====
		\$3,203,732	\$370,869,258	\$15,854,439

## Exercise 4:

Modify the program from exercise 3 so the report shows the maximum amounts granted to a state. The report should look like this:

```

                                NIH Fellowship, Research and Training Awards
                                Maximum Amounts Granted
Year  State      Fellowship      Research      Training
ffffffffff

1996  DIST OF COL    $246,684    $32,118,502    $623,425
      MARYLAND      $1,624,258  $181,608,047  $8,795,463
      VIRGINIA      $622,612    $41,466,312    $2,442,739
      =====
      $1,624,258    $181,608,047  $8,795,463

1997  DIST OF COL    $340,851    $33,651,458    $884,139
      MARYLAND      $1,476,525  $187,473,322  $9,478,438
      VIRGINIA      $618,832    $47,942,379    $2,783,142
      =====
      $1,476,525    $187,473,322  $9,478,438

```

## Exercise 5:

Now, create a report that includes both the total amounts granted and the maximums as shown here:

NIH Fellowship, Research and Training Awards Total and Maximum Amount Granted							
Year	State	Fellowship		Research		Training	
		Total	Maximum	Total	Maximum	Total	Maximum
1996	DIST OF COL	\$454,415	\$246,684	\$44,120,947	\$32,118,502	\$1,084,785	\$623,425
	MARYLAND	\$1,900,302	\$1,624,258	\$228,938,871	\$181,608,047	\$9,674,281	\$8,795,463
	VIRGINIA	\$892,258	\$622,612	\$78,893,975	\$41,466,312	\$3,696,527	\$2,442,739
		=====	=====	=====	=====	=====	=====
		\$3,246,975	\$1,624,258	\$351,953,793	\$181,608,047	\$14,455,593	\$8,795,463
1997	DIST OF COL	\$497,943	\$340,851	\$47,236,672	\$33,651,458	\$1,216,366	\$884,139
	MARYLAND	\$1,836,532	\$1,476,525	\$238,185,108	\$187,473,322	\$10,847,551	\$9,478,438
	VIRGINIA	\$869,257	\$618,832	\$85,447,478	\$47,942,379	\$3,790,522	\$2,783,142
		=====	=====	=====	=====	=====	=====
		\$3,203,732	\$1,476,525	\$370,869,258	\$187,473,322	\$15,854,439	\$9,478,438

# Across Variable

A column is created for each value of an across variable.

**Example 1:** In this example YEAR is an across variable.

```
proc report data=rep.awards nowd headskip headline;  
  column stname year, (resn resamt);  
  define year / ' ' across;  
  define stname / 'State' group width=11;  
  define resn / '/Number/Awards' ;  
  define resamt / '/Total /Amount' ;  
  format resamt dollar14. ;  
run;
```

# Across Variable

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

	1996		1997	
State	Number Awards	Total Amount	Number Awards	Total Amount
DIST OF COL	174	\$44,120,947	182	\$47,236,672
MARYLAND	781	\$228,938,871	817	\$238,185,108
VIRGINIA	337	\$78,893,975	357	\$85,447,478

# Across Variable

**Example 2:** In **sample 6** we assign a format to YEAR to add lines on each side of the year.

```
proc format;  
  value yr 1996='_1996_' 1997='_1997_';  
run;
```

```
proc report data=rep.awards nowd headskip headline;  
  column stname year, (resn resamt);  
  define year / ' ' across;  
  define stname / 'State' group width=11;  
  define resn / '/Number/Awards';  
  define resamt / '/Total/Amount';  
  format resamt dollar14. year yr.;  
run;
```

# Sample 6

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

	1996		1997	
State	Number Awards	Total Amount	Number Awards	Total Amount
DI ST OF COL	174	\$44, 120, 947	182	\$47, 236, 672
MARYLAND	781	\$228, 938, 871	817	\$238, 185, 108
VI RGI N I A	337	\$78, 893, 975	357	\$85, 447, 478



# Computed Variable

These are variables you create in PROC REPORT to display them in the report. You need to:

- include the variable in the COLUMN statement
- define the usage as COMPUTED
- use a COMPUTE block to compute the values

# Computed Variable

The variables you need to calculate a new variable must be included in the COLUMN statement even if you don't want to display them. If you don't want them displayed use the NOPRINT option in the corresponding DEFINE statements.

A computed variable can only depend on the variables to its left in the COLUMN statement.

# Computed Variable

## Example 1:

```
proc report data=rep.awards nowd headskip headline;
  column year stname uni vname conn fel n resn tran
           othn total;
  define year / 'Year' order;
  define stname / 'State' order;
  define uni vname / 'Uni versi ty' order;
  define conn / display noprnt;
  define fel n / display noprnt;
  define resn / display noprnt;
  define tran / display noprnt;
  define othn / display noprnt;
  define total / 'Number of/Awards' computed center
                width=9 format=4.;
  compute total;
    total=conn+fel n+resn+tran+othn;
  endcomp;
run;
```

# Computed Variable

Total NIH Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	University	Number of Awards
ff			
1996	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	35
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	159
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	19
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	681
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	250
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	21
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	215
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	165
1997	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	34
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	168
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	20
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	687
		U S UNIFORMED SERVICES UNIV HLTH SCI S	2
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	269
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	18
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	246
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	152

# Computed Variable

**Example 2:** In sample 7 the variable TOTAL is a computed variable.

```
proc report data=rep.awards nowd headskip;
  column year sname conn feln resn tran othn total;
  define year / 'Year' group width=4;
  define sname / 'State' group width=11;
  define conn / 'Contracts' width=11;
  define feln / 'Fellowships' width=11;
  define resn / 'Research' width=11;
  define tran / 'Training' width=11;
  define othn / 'Other' width=11;
  define total / 'Total' computed format=comma11.;
  break after year / skip summarize ol suppress;
  compute total;
    total=conn.sum + feln.sum + resn.sum + tran.sum + othn.sum;
  endcomp;
  format conn feln resn tran othn comma5.;
run;
```

# Computed Variable

The item **conn.sum** is called a **compound name**. A compound name is of the form: *variable.statistic*

It contains the statistic that PROC REPORT calculates for that variable.

# Sample 7

## NIH Awards Granted to Medical Schools District of Columbia, Maryland and Virginia

Year	State	Contracts	Fellowships	Research	Training	Other	Total
1996	DIST OF COL	14	18	174	6	1	213
	MARYLAND	25	68	781	54	3	931
	VIRGINIA	6	33	337	24	1	401
		ffffffffffff	ffffffffffff	ffffffffffff	ffffffffffff	ffffffffffff	ffffffffffff
		45	119	1,292	84	5	1,545
1997	DIST OF COL	13	19	182	8	0	222
	MARYLAND	21	63	817	54	3	958
	VIRGINIA	6	31	357	20	2	416
		ffffffffffff	ffffffffffff	ffffffffffff	ffffffffffff	ffffffffffff	ffffffffffff
		40	113	1,356	82	5	1,596

# Compute Blocks

A compute block contains one or more statements that are executed as the report is built. It starts with a COMPUTE statement and must end with an ENDCOMP statement.

You can include some DATA step statements in compute blocks:

assignment	DO	END
IF-THEN/ELSE	SELECT	sum



# Compute Blocks

There are two types of compute blocks:

**First type:** Associate the compute block with a column

```
COMPUTE report_item / type-specification;  
  statements  
  CALL DEFINE(column-id, ' attribute', value);  
ENDCOMP;
```

<i>report_item</i>	a data set variable, a computed variable, or a statistic to associate the block with
<i>type-specification</i>	computed variables are assumed to be numeric. If it is character you must use the option CHARACTER and, optionally, the LENGTH= option (default: 8).

# Compute Blocks

<i>column-id</i>	column name or column number. <code>_Cn_</code> indicates column number <i>n</i> . <code>_COL_</code> identifies the column that is attached to the compute block.
<i>attribute</i>	in the nonwindowing environment it can only be "format"
<i>value</i>	value for the attribute (e.g. comma11.)

# Compute Blocks

## Example 1:

```
proc report data=rep.awards nowd headskip;
  column year sname conn feln resn tran othn total s;
  define year / 'Year' group width=4;
  define sname / 'State' group width=11;
  define conn / 'Contracts' width=11;
  define feln / 'Fellowships' width=11;
  define resn / 'Research' width=11;
  define tran / 'Training' width=11;
  define othn / 'Other' width=11;
  define total / 'Total' computed width=11;
  define s / ' ' computed width=4;
  compute total;
    total=conn.sum + feln.sum + resn.sum +
          tran.sum + othn.sum;
    call define(_col_, "format", "comma11.");
  endcomp;
  compute s / character length=4;
    if total > 500 then s='>500';
    else s=' ';
  endcomp;
  break after year / skip;
  format conn feln resn tran othn comma5.;
run;
```

# Compute Blocks

NIH Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	Contracts	Fellowships	Research	Training	Other	Total	
1996	DIST OF COL	14	18	174	6	1	213	
	MARYLAND	25	68	781	54	3	931	>500
	VIRGINIA	6	33	337	24	1	401	
1997	DIST OF COL	13	19	182	8	0	222	
	MARYLAND	21	63	817	54	3	958	>500
	VIRGINIA	6	31	357	20	2	416	

# Compute Blocks

## Example 2:

```
proc report data=rep.awards nowd headskip headline;
  column stname year, (resn resamt)
           ('_Grand Total_' tresn tresamt);
  define year / ' ' across f=yr.;
  define stname / 'State' group width=11;
  define resn / '/Number/Awards' width=6 spacing=6;
  define resamt / '/Total/Amount' f=dollar12.;
  define tresn / '/Number/Awards' computed f=comma6. spacing=6;
  define tresamt / '/Total/Amount' computed f=dollar12.;
  compute tresn;
    tresn=_c2+_c4_;
  endcomp;
  compute tresamt;
    tresamt=_c3+_c5_;
  endcomp;
run;
```

# Compute Blocks

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

	_____1996_____		_____1997_____		_____Grand Total_____	
State	Number Awards	Total Amount	Number Awards	Total Amount	Number Awards	Total Amount
DIST OF COL	174	\$44,120,947	182	\$47,236,672	356	\$91,357,619
MARYLAND	781	\$228,938,871	817	\$238,185,108	1,598	\$467,123,979
VIRGINIA	337	\$78,893,975	357	\$85,447,478	694	\$164,341,453

# Compute Blocks

**Second type:** Associate the compute block with a location and, optionally, a break variable

```
COMPUTE location <break-var>;  
  LINE specifications;  
  other statements;  
ENDCOMP;
```

<i>location</i>	where compute block executes: AFTER, BEFORE
<i>break-var</i>	(optional) a group or order variable

# Compute Blocks

**Example:** This is a modified version of sample 4. Here we use a COMPUTE AFTER.

```
proc report data=rep.awards nowd headskip headline;  
  column year sname resn resamt;  
  define year / 'Year' group;  
  define sname / 'State' group;  
  define resn / 'Number/Awards';  
  define resamt / 'Total/Amount';  
  break after year / skip;  
  compute after;  
    line 'The total amount awarded from 1996 to 1997 is'  
      resamt.sum dollar13. '.';  
  endcomp;  
  format resamt dollar13.;  
run;
```

Note: The LINE statement will be covered in the next section.



# Compute Blocks

NIH Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	Number Awards	Total Amount
1996	DIST OF COL	174	\$44, 120, 947
	MARYLAND	781	\$228, 938, 871
	VIRGINIA	337	\$78, 893, 975
1997	DIST OF COL	182	\$47, 236, 672
	MARYLAND	817	\$238, 185, 108
	VIRGINIA	357	\$85, 447, 478

The total amount awarded from 1996 to 1997 is \$722,823,051.

# LINE Statement

Is a subset of the PUT statement. It can only be used with compute blocks that are associated with a location in the report.

It cannot be used in conditional statements.

**LINE** *specifications*;

*specifications* can be any of the following:

<i>item</i>	data set variable, computed variable, statistic in the report
<i>format</i>	format for variable
<i>'character string'</i>	string of text
<i>@n</i>	specifies the number of the column in which to start writing the item following it
<i>+n</i>	specifies the number of columns to skip before writing the next item

# **Workshop 2**

## Exercise 1:

Create a report that shows the total amount granted for Training Awards for each state for each year. The year should be shown across the page. The report would look like this:

NIH Training Awards Amounts Granted		
State	Year	
	1996	1997
////////////////////////////////////		
DIST OF COL	\$1,084,785	\$1,216,366
MARYLAND	\$9,674,281	\$10,847,551
VIRGINIA	\$3,696,527	\$3,790,522

## Exercise 2:

Modify the program from exercise 1 so it also includes a column for the percent increase from 1996 to 1997. Here is the report:

NIH Training Awards Amounts Granted			
State	Year		Percent Increase
	1996	1997	
////////////////////////////////////			
DIST OF COL	\$1,084,785	\$1,216,366	12.13%
MARYLAND	\$9,674,281	\$10,847,551	12.13%
VIRGINIA	\$3,696,527	\$3,790,522	2.54%

**Note:** To display percent signs use the PERCENTw.d format where "w" represents the total width and "d" is the number of decimal digits (0 or 2). When you use this format the value is first multiplied by 100. Then the value is formatted using the w.d format and a % sign is added to the end. If the value is negative, parenthesis are added. Allow enough room for all the digits, the % sign and the parenthesis even if the value is not negative. For example, to display the value 100.00% use the format PERCENT9.2.

## Exercise 3:

Modify the program from exercise 1 in workshop 1 so it includes a column for the total amount granted for Fellowship, Research and Training awards.

NIH Fellowship, Research and Training Awards						
Year	State	University	Fellowship	Research	Training	Total
1996	DIST OF COL	GEORGE WASHINGTON UNIV SCH OF MEDICINE	\$136,783	\$7,097,448	\$0	\$7,234,231
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	\$246,684	\$32,118,502	\$461,360	\$32,826,546
		HOWARD UNIVERSITY COLLEGE OF MEDICINE	\$70,948	\$4,904,997	\$623,425	\$5,599,370
	MARYLAND	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	\$1,624,258	\$181,608,047	\$8,795,463	\$192,027,768
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	\$276,044	\$47,330,824	\$878,818	\$48,485,686
	VIRGINIA	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	\$48,626	\$3,472,404	\$0	\$3,521,030
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	\$622,612	\$41,466,312	\$2,442,739	\$44,531,663
1997	DIST OF COL	VIRGINIA COMMONWEALTH UNIV MED COL OF VA	\$221,020	\$33,955,259	\$1,253,788	\$35,430,067
		GEORGE WASHINGTON UNIV SCH OF MEDICINE	\$126,803	\$7,015,966	\$0	\$7,142,769
		GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	\$340,851	\$33,651,458	\$884,139	\$34,876,448
	MARYLAND	HOWARD UNIVERSITY COLLEGE OF MEDICINE	\$30,289	\$6,569,248	\$332,227	\$6,931,764
		JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	\$1,476,525	\$187,473,322	\$9,478,438	\$198,428,285
	VIRGINIA	U S UNIFORMED SERVICES UNIV HLTH SCI S	\$0	\$97,773	\$0	\$97,773
		UNIVERSITY OF MARYLAND SCH OF MEDICINE	\$360,007	\$50,614,013	\$1,369,113	\$52,343,133
		EASTERN VIRGINIA MED/SCH MED COL HAMP RD	\$16,718	\$3,248,901	\$0	\$3,265,619
		UNIVERSITY OF VIRGINIA SCH OF MEDICINE	\$618,832	\$47,942,379	\$2,783,142	\$51,344,353
		VIRGINIA COMMONWEALTH UNIV MED COL OF VA	\$233,707	\$34,256,198	\$1,007,380	\$35,497,285

# Missing Data

By default, PROC REPORT ignores observations with a missing value for any group, order or across variables that appear in the report.

To consider missing values as valid values use the MISSING option in the PROC REPORT statement.

Here we add one observation where the state name is missing.

```
data awmi ss;  
  set rep.awards end=e; output;  
  if e=1 then do;  
    year=1996; stname=' '; resn=20; resamt=500000;  
    output;  
  end;  
run;
```

# Missing Data

```
*without missing option;  
proc report data=awmiss nowd headskip headline;  
  column year stname resn resamt;  
  define year / 'Year' group;  
  define stname / 'State' group;  
  define resn / 'Number/Awards' f=comma6. ;  
  define resamt / 'Total/Amount' f=dollar13. ;  
  break after year / skip summarize suppress ol ;  
  rbreak after / dol summarize;  
run;
```

```
*with missing option;  
proc report data=awmiss nowd headskip headline missing;  
  column year stname resn resamt;  
  define year / 'Year' group;  
  define stname / 'State' group;  
  define resn / 'Number/Awards' f=comma6. ;  
  define resamt / 'Total/Amount' f=dollar13. ;  
  break after year / skip summarize suppress ol ;  
  rbreak after / dol summarize;  
run;
```



# Missing Data

## Without MISSING Option

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	Number Awards	Total Amount
~~~~~			
1996	DIST OF COL	174	\$44,120,947
	MARYLAND	781	\$228,938,871
	VIRGINIA	337	\$78,893,975
		~~~~~	~~~~~
		1,292	\$351,953,793
1997	DIST OF COL	182	\$47,236,672
	MARYLAND	817	\$238,185,108
	VIRGINIA	357	\$85,447,478
		~~~~~	~~~~~
		1,356	\$370,869,258
		=====	=====
		2,648	\$722,823,051

## With MISSING Option

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	Number Awards	Total Amount
~~~~~			
1996		20	\$500,000
	DIST OF COL	174	\$44,120,947
	MARYLAND	781	\$228,938,871
	VIRGINIA	337	\$78,893,975
		~~~~~	~~~~~
		1,312	\$352,453,793
1997	DIST OF COL	182	\$47,236,672
	MARYLAND	817	\$238,185,108
	VIRGINIA	357	\$85,447,478
		~~~~~	~~~~~
		1,356	\$370,869,258
		=====	=====
		2,668	\$723,323,051

# How a Report is Built

1. If necessary, the data is consolidated by group, order and across variables.
2. All report statistics are calculated, including those for the summary lines. All this information is stored in a temporary file.
3. All **DATA step variables** are initialized to missing. These are variables that are created in compute blocks but not displayed in the report.
4. One at a time, each row is built and displayed:
  - a. **report variables** (all variables that are displayed in the report) are initialized to missing.

# How a Report is Built

- b. values are filled in from left to right.  
Computed variables are calculated using their respective compute blocks. Other values are obtained from the temporary file that was created at the beginning of the process.
- c. at a break, PROC REPORT builds the break lines created with the BREAK or RBREAK statements then executes their respective compute blocks if any exist.

# How a Report is Built

Suppose you want a report like the one below with a percent column that shows the percentage of research awards granted to a state for each year separately.

NIH Research Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	N	Percent
ff			
1996	DIST OF COL	174	13.47%
	MARYLAND	781	60.45%
	VIRGINIA	337	26.08%
		fffff	ffffff
		1,292	100.00%
1997	DIST OF COL	182	13.42%
	MARYLAND	817	60.25%
	VIRGINIA	357	26.33%
		fffff	ffffff
		1,356	100.00%

# How a Report is Built

This program will display that report. An explanation follows in the next pages.

```
proc report data=rep.awards nowd headskip headline;  
  column year sname resn respct;  
  define year / 'Year' group;  
  define sname / 'State' group;  
  define resn / 'N' f=comma5.;  
  define respct / 'Percent' computed center f=percent9.2;  
  break after year / skip summarize ol suppress;  
  compute before year;  
    resntot=resn.sum;  
  endcomp;  
  compute respct;  
    respct=resn.sum/resntot;  
  endcomp;  
run;
```

# How a Report is Built

In the program we create a variable called RESPCT that will contain the percentage. In the COMPUTE RESPCT block we need to divide the total number of research awards for a state by the total number of research awards for that year.

It is important to note that the value of RESN.SUM has different meanings in different lines of the report. In detail lines it represents the number of research awards for that state for that year. In the summary line it represents the number of research awards for that year. These are the two values we need to divide to obtain the percentage.

# How a Report is Built

We need to retain the value of RESN.SUM for the year **before** PROC REPORT calculates each detail line since that is the denominator for calculating RESPCT. To achieve this we use the COMPUTE BEFORE YEAR block which retrieves the value of RESN.SUM for a year and saves it in the DATA step variable RESNTOT.

Later, when PROC REPORT calculates the values of RESPCT, the variable RESNTOT is used as the denominator.

# Modifying the Data to Get the Report you Want

Suppose you want a report like the one below for each year. Can it be done with the data set as it is saved?

```

NIH Awards Granted to Medical Schools
District of Columbia, Maryland and Virginia

Year State      Award Type      N      Amount
#####
1996 DIST OF COL Research      174    44,120,947
      Training        6      1,084,785
      Contracts      14      6,568,339
      Fellowships    18      454,415
      Other           1       734,544
      ffffff ffffffff
      213    $52,963,030

      MARYLAND      Research      781    228,938,871
      Training        54      9,674,281
      Contracts      25     14,249,751
      Fellowships    68      1,900,302
      Other           3       1,574,112
      ffffff ffffffff
      931    $256,337,317

      VIRGINIA      Research      337     78,893,975
      Training        24      3,696,527
      Contracts        6      1,649,995
      Fellowships    33       892,258
      Other           1       132,978
      ffffff ffffffff
      401    $85,265,733

      =====
      1,545 $394,566,080
      =====
  
```



# Modifying the Data to Get the Report you Want

The format of the data set needs to be changed. The following DATA step creates the data set TYPES that is saved differently. For each observation in AWARDS it creates five observations, one for each type of award.

```
data types;
  set rep.awards;
  array amtaward(5) resamt traamt conamt felamt othamt;
  array naward(5) resn tran conn feln othn;
  do type=1 to 5;
    amount=amtaward(type);
    num=naward(type);
    output;
  end;
  keep stname year type amount num;
run;
```

# Modifying the Data to Get the Report you Want

First 20 observations of data set TYPES

Obs	STNAME	YEAR	TYPE	AMOUNT	NUM
1	MARYLAND	1996	1	181608047	560
2	MARYLAND	1996	2	8795463	44
3	MARYLAND	1996	3	10304633	17
4	MARYLAND	1996	4	1624258	57
5	MARYLAND	1996	5	1574112	3
6	MARYLAND	1996	1	47330824	221
7	MARYLAND	1996	2	878818	10
8	MARYLAND	1996	3	3945118	8
9	MARYLAND	1996	4	276044	11
10	MARYLAND	1996	5	0	0
11	VI RGI NI A	1996	1	41466312	178
12	VI RGI NI A	1996	2	2442739	14
13	VI RGI NI A	1996	3	0	0
14	VI RGI NI A	1996	4	622612	22
15	VI RGI NI A	1996	5	132978	1
16	DI ST OF COL	1996	1	32118502	134
17	DI ST OF COL	1996	2	461360	4
18	DI ST OF COL	1996	3	5288387	11
19	DI ST OF COL	1996	4	246684	10
20	DI ST OF COL	1996	5	0	0

Compare to the original data set. 

# Modifying the Data to Get the Report you Want

```
proc format;
  value atype 1=' Research' 2=' Training'
              3=' Contracts' 4=' Fellowshi ps' 5=' Other' ;
run;

proc report data=types nowd headline headskip;
  column year sname type num amount;
  define year / 'Year' group;
  define sname / 'State' group;
  define type / 'Award Type' group order=internal f=atype.;
  define num / 'N' analysis sum f= comma5.;
  define amount / 'Amount' f=comma12.;
  break after sname / skip;
  break after year / page summarize suppress dul dol;
  compute amount;
    if year=. and sname=' ' and type=. then
      call define(_col_, "format", "dollar12.");
  endcomp;
run;
```

# Multiple Panels per Page

The PANELS= option produces several panels per page as shown below. The option is used in the PROC REPORT statement.

NIH Awards Granted to Medical Schools District of Columbia, Maryland and Virginia							
Year	State	Award Type	N	Year	State	Award Type	N
////////////////////////////////////				////////////////////////////////////			
1996	DIST OF COL	Research	174	1997	DIST OF COL	Research	182
		Training	6			Training	8
		Contracts	14			Contracts	13
		Fellowships	18			Fellowships	19
		Other	1			Other	0
	MARYLAND	Research	781		MARYLAND	Research	817
		Training	54			Training	54
		Contracts	25			Contracts	21
		Fellowships	68			Fellowships	63
		Other	3			Other	3
	VIRGINIA	Research	337		VIRGINIA	Research	357
		Training	24			Training	20
		Contracts	6			Contracts	6
		Fellowships	33			Fellowships	31
		Other	1			Other	2

# Multiple Panels per Page

```
proc report data=types nowd headline headskip  
           panels=2 ps=21;  
  column year stname type num;  
  define year    / 'Year' group;  
  define stname  / 'State' group;  
  define type    / 'Award Type' group order=internal;  
  define num     / 'N' analysis sum;  
  format type atype.;  
run;
```

# Displaying Computed Variable Before Variables it Uses

Computed variables can only depend on variables to the left of it. Here is an example of what you can do to override this restriction.

```
proc report data=rep.awards nowd headskip headline;  
  column year uni vname total conn fel n resn tran othn;  
  define year / 'Year' group width=4;  
  define uni vname / 'University' group;  
  define conn / 'Contracts' width=11;  
  define fel n / 'Fellowships' width=11;  
  define resn / 'Research' width=11;  
  define tran / 'Training' width=11;
```

(continued on next page)

# Displaying Computed Variable Before Variables it Uses

(continuation)

```
define othn / 'Other' width=11;  
define total / 'Number of/Awards' computed center  
              width=9 format=4. ;  
compute before uni vname;  
    tot=conn.sum+fel n.sum+resn.sum+tran.sum+othn.sum;  
endcomp;  
compute total ;  
    total=tot;  
endcomp;  
break after year / skip;  
run;
```

# Displaying Computed Variable Before Variables it Uses

NIH Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	Uni versi ty	Number of Awards	Contracts	Fel lowshi ps	Research	Trai ni ng	Other
<div style="border: 1px solid black; height: 15px; width: 100%;"></div>							
1996	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	21	1	2	18	0	0
	GEORGE WASHINGTON UNIV SCH OF MEDICINE	35	2	6	26	0	1
	GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	159	11	10	134	4	0
	HOWARD UNIVERSITY COLLEGE OF MEDICINE	19	1	2	14	2	0
	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	681	17	57	560	44	3
	UNIVERSITY OF MARYLAND SCH OF MEDICINE	250	8	11	221	10	0
	UNIVERSITY OF VIRGINIA SCH OF MEDICINE	215	0	22	178	14	1
	VIRGINIA COMMONWEALTH UNIV MED COL OF VA	165	5	9	141	10	0
1997	EASTERN VIRGINIA MED/SCH MED COL HAMP RD	18	1	1	16	0	0
	GEORGE WASHINGTON UNIV SCH OF MEDICINE	34	3	4	27	0	0
	GEORGETOWN UNIVERSITY SCHOOL OF MEDICINE	168	8	14	140	6	0
	HOWARD UNIVERSITY COLLEGE OF MEDICINE	20	2	1	15	2	0
	JOHNS HOPKINS UNIVERSITY SCH OF MEDICINE	687	15	50	579	42	1
	U S UNIFORMED SERVICES UNIV HLTH SCI S	2	0	0	1	0	1
	UNIVERSITY OF MARYLAND SCH OF MEDICINE	269	6	13	237	12	1
	UNIVERSITY OF VIRGINIA SCH OF MEDICINE	246	0	22	208	14	2
	VIRGINIA COMMONWEALTH UNIV MED COL OF VA	152	5	8	133	6	0



# **Workshop 3**

## Exercise 1:

Create the following report. You will need to modify the data like we did in class. Calculate the column "Average" in the PROC REPORT. Verify that the average values are correct.

NIH Awards Granted to Medical Schools District of Columbia, Maryland and Virginia				
Year	Award Type	N	Total	Average
ff				
1996	Research	1,292	351,953,793	272,410
	Training	84	14,455,593	172,090
	Contracts	45	22,468,085	499,291
	Fellowships	119	3,246,975	27,286
	Other	5	2,441,634	488,327
1997	Research	1,356	370,869,258	273,502
	Training	82	15,854,439	193,347
	Contracts	40	23,097,510	577,438
	Fellowships	113	3,203,732	28,352
	Other	5	729,383	145,877

## Exercise 2:

Modify exercise 1 so it also includes a column that illustrates the percentage granted to each type of award for each year separately. Here is the report:

NIH Awards Granted to Medical Schools District of Columbia, Maryland and Virginia					
Year	Award Type	N	Total	Average	Percent
<div style="border: 1px solid black; height: 15px; width: 100%;"></div>					
1996	Research	1,292	351,953,793	272,410	89.20
	Training	84	14,455,593	172,090	3.66
	Contracts	45	22,468,085	499,291	5.69
	Fellowships	119	3,246,975	27,286	0.82
	Other	5	2,441,634	488,327	0.62
1997	Research	1,356	370,869,258	273,502	89.64
	Training	82	15,854,439	193,347	3.83
	Contracts	40	23,097,510	577,438	5.58
	Fellowships	113	3,203,732	28,352	0.77
	Other	5	729,383	145,877	0.18

## Exercise 3:

Modify exercise 1 so the Average column is displayed before the N and Total columns.

```
NIH Awards Granted to Medical Schools
District of Columbia, Maryland and Virginia

Year   Award Type   Average   N       Total
fffffffffffffffffffffffffffffffffffffffffffffffffffffffffffff

1996   Research      272,410   1,292   351,953,793
      Training      172,090     84     14,455,593
      Contracts     499,291     45     22,468,085
      Fellowships    27,286    119     3,246,975
      Other         488,327     5      2,441,634

1997   Research      273,502   1,356   370,869,258
      Training      193,347     82     15,854,439
      Contracts     577,438     40     23,097,510
      Fellowships    28,352    113     3,203,732
      Other         145,877     5      729,383
```

# Saving a Report Definition

You can save a report definition by using the OUTREPT= option in the PROC REPORT statement. To suppress the display of the report use the NOEXEC option also.

The OUTREPT= option specifies the catalog entry where the report definition will be saved. It takes a three part name:

*libref.catname.entry\_name*

<i>libref</i>	libref specified in LIBNAME statement
<i>catname</i>	catalog name
<i>entry_name</i>	an entry name, i.e. the name for the report definition

# Saving a Report Definition

Here we modify sample 7 to save the report definition permanently:

```
libname repdef 'c:\awards';

title 'NIH Awards Granted to Medical Schools';
title2 'District of Columbia, Maryland and Virginia';

proc report data=rep.awards nowd headskip headline
            outrept=repdef.reports.report1 noexec;
  column year sname conn feln resn tran othn total;
  define year / 'Year' group width=4;
  define sname / 'State' group width=11;
  define conn / 'Contracts' width=11;
  define feln / 'Fellowships' width=11;
  define resn / 'Research' width=11;
  define tran / 'Training' width=11;
  define othn / 'Other' width=11;
  define total / 'Total' computed width=11 f=comma11.;
  break after year / skip summarize ol suppress;
  compute total;
    total =conn.sum+feln.sum+resn.sum+tran.sum+othn.sum;
  endcomp;
  format conn feln resn tran othn comma5.;
run;
```

# Using a Report Definition

To use a report definition use the REPORT= option in the PROC REPORT statement to point to the entry containing the report.

To view the statements that creates the report use the LIST option in the PROC statement. The statements will be displayed in the SAS log.

```
libname repdef 'c:\awards' ;

title  '1998 NIH Awards Granted to Medical Schools' ;
title2 'District of Columbia, Maryland and Virginia' ;

proc report data=rep.awards98 nowd
            report=repdef.reports.report1;
run;
```

# Using a Report Definition

1998 NIH Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Year	State	Contracts	Fellowships	Research	Training	Other	Total
////////////////////////////////////							
1998	DIST OF COL	15	19	173	6	0	213
	MARYLAND	22	53	824	53	4	956
	VIRGINIA	5	35	381	22	2	445
		////////////////////////////////	////////////////////////////////	////////////////////////////////	////////////////////////////////	////////////////////////////////	////////////////////////////////
		42	107	1,378	81	6	1,614



# Saving the Results in a SAS Data Set

You can save the results of the procedure in a SAS data set by using the OUT= option in the PROC REPORT statement.

```
proc report data=rep.awards nowd headskip headline  
            out=rep.report1;  
  column year sname conn fel n resn tran othn total;  
  define year / 'Year' group width=4;  
  define sname / 'State' group width=11;  
  define conn / 'Contracts' width=11;  
  define fel n / 'Fellowships' width=11;  
  define resn / 'Research' width=11;  
  define tran / 'Training' width=11;  
  define othn / 'Other' width=11;  
  define total / 'Total' computed width=11 f=comma11.;  
  break after year / skip summarize ol suppress;  
  compute total;  
    total =conn.sum+fel n.sum+resn.sum+tran.sum+othn.sum;  
  endcomp;  
  format conn fel n resn tran othn comma5.;  
run;
```

# Saving the Results in a SAS Data Set

This is what the SAS data set REPORT1 looks like:

YEAR	STNAME	CONN	FELN	RESN	TRAN	OTHN	TOTAL	_BREAK_
1996	DI ST OF COL	14	18	174	6	1	213	
1996	MARYLAND	25	68	781	54	3	931	
1996	VI RGI NI A	6	33	337	24	1	401	
1996		45	119	1, 292	84	5	1545	YEAR
1997	DI ST OF COL	13	19	182	8	0	222	
1997	MARYLAND	21	63	817	54	3	958	
1997	VI RGI NI A	6	31	357	20	2	416	
1997		40	113	1, 356	82	5	1596	YEAR

# The **\_PAGE\_** Option (V7 & above)

The **\_PAGE\_** option can be used in compute blocks that are associated with a location. It causes the compute block to run once for each page, either immediately after printing titles or immediately before printing footnotes.

COMPUTE *location* **\_PAGE\_** / *justification*;

<i>location</i>	BEFORE or AFTER
<i>justification</i>	LEFT, CENTER or RIGHT

# The **\_PAGE\_** Option (V7 & above)

```
proc report data=types nowd headline headskip;  
  column year sname type num;  
  define year      / group noprint;  
  define sname     / 'State' group;  
  define type      / 'Award Type' group order=internal;  
  define num       / 'N' f=comma5.;  
  break after sname / skip;  
  break after year / page summarize dul dol suppress;  
  compute before _page_ / center;  
    line 'Number of Awards for Year ' year 4.;  
    line ' ';  
  endcomp;  
  format type atype.;  
run;
```

# The \_PAGE\_ Option (V7 & above)

NIH Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Number of Awards for Year 1996

State	Award Type	N
ffffffffffffffffffffffffffffffff		
DIST OF COL	Research	174
	Training	6
	Fellowships	18
	Contracts	14
	Other	1
MARYLAND	Research	781
	Training	54
	Fellowships	68
	Contracts	25
	Other	3
VIRGINIA	Research	337
	Training	24
	Fellowships	33
	Contracts	6
	Other	1
		=====
		1,545
		=====

# The \_PAGE\_ Option (V7 & above)

NIH Awards Granted to Medical Schools  
District of Columbia, Maryland and Virginia

Number of Awards for Year 1997

State	Award Type	N
ff		

DIST OF COL	Research	182
	Training	8
	Fellowships	19
	Contracts	13
	Other	0

MARYLAND	Research	817
	Training	54
	Fellowships	63
	Contracts	21
	Other	3

VIRGINIA	Research	357
	Training	20
	Fellowships	31
	Contracts	6
	Other	2

=====  
1,596  
=====

# Output Delivery System (V7 & above)

A new feature in version 7 that is available for all procedures is the output delivery system, or ODS. It allows you to save SAS output in many formats.

To save the results of PROC REPORT in an HTML file, include an ODS HTML statement with the BODY= option before invoking PROC REPORT. After the RUN statement add the ODS HTML statement with the CLOSE option.

You could save the reports produced by PROC REPORT in Excel by creating an HTML file first as described above then reading it with Excel.

# Output Delivery System (V7 & above)


```
ods html body='c:\awards\res_awards.html';

title 'NIH Research Awards Granted to Medical Schools';
title2 'District of Columbia, Maryland and Virginia';
proc report data=rep.awards nowd headskip headline;
  column year stname resn resamt;
  define year / 'Year' group;
  define stname / 'State' group;
  define resn / 'N' f=comma5.;
  define resamt / 'Total' f=dollar12.;
  break after year / skip summarize ol suppress;
run;

ods html close;
```



# Output Delivery System (V7 & above)



The screenshot shows a web browser window with the title "C:\awards\html\res\_awards.html - Microsoft Internet Expl...". The address bar shows the file path. The menu bar includes File, Edit, View, Favorites, Tools, and Help. The main content area displays the title "NIH Research Awards Granted to Medical Schools District of Columbia, Maryland and Virginia" in a bold, italicized font. Below the title is a table with four columns: Year, State, N, and Total. The table contains data for the years 1996 and 1997, listing awards for the District of Columbia, Maryland, and Virginia, along with a subtotal for each year.

Year	State	N	Total
1996	DIST OF COL	174	\$44,120,947
	MARYLAND	781	\$228,938,871
	VIRGINIA	337	\$78,893,975
		1,292	\$351,953,793
1997	DIST OF COL	182	\$47,236,672
	MARYLAND	817	\$238,185,108
	VIRGINIA	357	\$85,447,478
		1,356	\$370,869,258

# **Workshop 4**

## Exercise 1:

Modify exercise 5 from workshop 1 so the report definition is saved permanently in the entry:

```
rep.reports.report1
```

## Exercise 2:

Use the report definition created above to obtain a report for year 1998. The set for 1998 is called **AWARDS98**. Here is the report:

```
NIH Fellowship, Research and Training Awards
Total and Average Amount Granted

Year: 1998
```

Year	State	Fellowship		Research		Training	
		Total	Maximum	Total	Maximum	Total	Maximum
1998	DIST OF COL	\$433,986	\$335,357	\$51,886,757	\$36,743,638	\$1,004,978	\$763,989
	MARYLAND	\$1,655,884	\$1,324,581	\$254,942,411	\$200,715,108	\$11,106,154	\$9,665,239
	VIRGINIA	\$1,026,209	\$638,819	\$94,181,809	\$55,229,701	\$3,628,904	\$2,523,083
		=====	=====	=====	=====	=====	=====
		\$3,116,079	\$1,324,581	\$401,010,977	\$200,715,108	\$15,740,036	\$9,665,239

# **Appendix: Workshop Answers**

```

*Enter these statements before running programs;
libname rep 'c:\awards'; *classroom;
options ls=144 ps=45 nonumber nodate; *8pt,lands, 0.5" margins;

```

```

/*-----
| WORKSHOP 1                                     |
-----*/

```

```

*Exercise 1;
title 'NIH Fellowship, Research and Training Awards';
proc report data=rep.awards nowd headskip headline;
  column year stname univname felamt resamt traamt;
  define year / 'Year' order;
  define stname / 'State' order;
  define univname / 'University' order;
  define felamt / 'Fellowship' format=dollar14.;
  define resamt / 'Research' format=dollar14.;
  define traamt / 'Training' format=dollar14.;
run;

```

```

*Exercise 2;
title 'NIH Fellowship, Research and Training Awards';
proc report data=rep.awards nowd headskip headline;
  column year stname univname felamt resamt traamt;
  define year / 'Year' order;
  define stname / 'State' order;
  define univname / 'University' order;
  define felamt / 'Fellowship' format=dollar14.;
  define resamt / 'Research' format=dollar14.;
  define traamt / 'Training' format=dollar14.;
  break after stname / skip summarize suppress ol;
  break after year / summarize suppress dol dul page;
run;

```

```

*Exercise 3;
title  'NIH Fellowship, Research and Training Awards';
proc report data=rep.awards nowd headskip headline;
  column year stname felamt resamt traamt;
  define year / 'Year' group;
  define stname / 'State' group;
  define felamt / 'Fellowship' format=dollar14.;
  define resamt / 'Research' format=dollar14.;
  define traamt / 'Training' format=dollar14.;
  break after year / skip summarize suppress dol;
run;

*Exercise 4;
title  'NIH Fellowship, Research and Training Awards';
title2 'Maximum Amounts Granted';
proc report data=rep.awards nowd headskip headline;
  column year stname felamt resamt traamt;
  define year / 'Year' group;
  define stname / 'State' group;
  define felamt / 'Fellowship' max format=dollar12.;
  define resamt / 'Research' max format=dollar12.;
  define traamt / 'Training' max format=dollar12.;
  break after year / skip summarize suppress dol;
run;

*Exercise 5;
title  'NIH Fellowship, Research and Training Awards';
title2 'Total and Maximum Amount Granted';
proc report data=rep.awards nowd headskip headline;
  column year stname felamt,(sum max) resamt,(sum max) traamt,(sum max);
  define year / 'Year' group;
  define stname / 'State' group;
  define felamt / 'Fellowship/--';
  define resamt / 'Research/--';
  define traamt / 'Training/--';
  define sum / 'Total' format=dollar14.;
  define max / 'Maximum' format=dollar14.;
  break after year / skip summarize suppress dol;
run;

```

```

/*-----
| WORKSHOP 2
|-----*/

*Exercise 1;
title 'NIH Training Awards';
title2 'Amounts Granted';
proc report data=rep.awards nowd headskip headline;
  column stname year,traamt;
  define year / 'Year/ ' across;
  define stname / 'State/ ' group;
  define traamt / ' ' format=dollar14. center;
run;

*Exercise 2;
title 'NIH Training Awards';
title2 'Amounts Granted';
proc report data=rep.awards nowd headskip headline;
  column stname year,traamt pctinc;
  define year / 'Year/ ' across;
  define stname / 'State/ ' group;
  define traamt / ' ' format=dollar14. center;
  define pctinc / 'Percent/Increase/ ' computed f=percent8.2;
  compute pctinc;
    pctinc=(_c3_-_c2_)/_c2_;
  endcomp;
run;

```

```

*Exercise 3;
*solution 1: use FELAMT, RESAMT, TRAAMT as analysis vars (default);
title 'NIH Fellowship, Research and Training Awards';
proc report data=rep.awards nowd headskip headline;
  column year stname univname felamt resamt traamt total;
  define year / 'Year' order;
  define stname / 'State' order;
  define univname / 'University' order;
  define felamt / 'Fellowship' format=dollar14.;
  define resamt / 'Research' format=dollar14.;
  define traamt / 'Training' format=dollar14.;
  define total / 'Total' computed format=dollar14.;
  compute total;
    total=felamt.sum+resamt.sum+traamt.sum;
  endcomp;
run;

*solution 2: use FELAMT, RESAMT, TRAAMT as display vars;
title 'NIH Fellowship, Research and Training Awards';
proc report data=rep.awards nowd headskip headline;
  column year stname univname felamt resamt traamt total;
  define year / 'Year' order;
  define stname / 'State' order;
  define univname / 'University' order;
  define felamt / 'Fellowship' format=dollar14. display;
  define resamt / 'Research' format=dollar14. display;
  define traamt / 'Training' format=dollar14. display;
  define total / 'Total' computed format=dollar14.;
  compute total;
    total=felamt+resamt+traamt;
  endcomp;
run;

```



```

/*-----
| WORKSHOP 3
|-----*/
*Exercise 1;
data types;
  set rep.awards;
  array amtaward(5) resamt traamt conamt felamt othamt;
  array naward(5) resn tran conn feln othn;
  do type=1 to 5;
    amount=amtaward(type);
    num=naward(type);
    output;
  end;
  keep sname year type amount num;
run;

proc format;
  value atype 1='Research' 2='Training' 3='Contracts'
              4='Fellowships' 5='Other';
run;

proc report data=types nowd headline headskip;
  column year type num amount avgamt;
  define year    / 'Year' group;
  define type    / 'Award Type' group order=internal f=atype.;
  define num      / 'N' analysis sum center f=comma5.;
  define amount  / 'Total' f=comma12. center;
  define avgamt  / 'Average' computed f=comma8. center;
  break after year / skip;
  compute avgamt;
    avgamt=amount.sum/num.sum;
  endcomp;
run;

```

```

*Exercise 2;
title  'NIH Awards Granted to Medical Schools';
title2 'District of Columbia, Maryland and Virginia';
proc report data=types nowd headline headskip;
  column year type num amount avgamt pct;
  define year    / 'Year' group;
  define type    / 'Award Type' group order=internal f=atype.;
  define num      / 'N' analysis sum center f=comma5.;
  define amount  / 'Total' f=comma12. center;
  define avgamt  / 'Average' computed f=comma8. center;
  define pct     / 'Percent' computed f=6.2 width=7;
  break after year / skip;
  compute avgamt;
    avgamt=amount.sum/num.sum;
  endcomp;
  compute before year;
    yrtot=amount.sum;
  endcomp;
  compute pct;
    pct=100*amount.sum/yrtot;
  endcomp;
run;

```

```

*Exercise 3;
title  'NIH Awards Granted to Medical Schools';
title2 'District of Columbia, Maryland and Virginia';
proc report data=types nowd headline headskip;
  column year type avgamt num amount;
  define year    / 'Year' group;
  define type    / 'Award Type' group order=internal f=atype.;
  define num     / 'N' analysis sum center f=comma5.;
  define amount  / 'Total' f=comma12. center;
  define avgamt  / 'Average' computed f=comma8. center;
  break after year / skip;
  compute before type;
    avg=amount.sum/num.sum;
  endcomp;
  compute avgamt;
    avgamt=avg;
  endcomp;
run;

```

```

/*-----
| WORKSHOP 4                                     |
|-----*/
*Exercise 1;
title  'NIH Fellowship, Research and Training Awards';
title2 'Total and Maximum Amount Granted';
proc report data=rep.awards nowd headskip headline outrept=rep.gr_rept.stats noexec;
  column year sname felamt,(sum max) resamt,(sum max) traamt,(sum max);
  define year / 'Year' group;
  define sname / 'State' group;
  define felamt / 'Fellowship/--';
  define resamt / 'Research/--';
  define traamt / 'Training/--';
  define sum / 'Total' format=dollar14.;
  define max / 'Maximum' format=dollar14.;
  break after year / skip summarize suppress dol;
run;

*Exercise 2;
title  'NIH Fellowship, Research and Training Awards';
title2 'Total and Average Amount Granted';
title4 'Year: 1998';
proc report data=rep.awards98 nowd report=rep.gr_rept.stats;
run;

```

